######################################################
##
## EXERCISE 10
##
######################################################

## List all persons with a birthdate and a death date, and in the second column, show the number of years that the person lived.
#
SELECT ?person ?approxAge
WHERE {
  ?person schema:birthDate ?birth ;
          schema:deathDate ?death;
  .
  BIND ( year(?death) - year(?birth) AS ?approxAge )
} 

## Query that outputs a string consisting of a person’s given name, family name and honorific suffix, separated by spaces
#
SELECT ?output 
WHERE {
  ?person schema:givenName ?givenName ;
          schema:familyName ?familyName ;
          schema:honorificSuffix ?suffix ;
  .
  BIND ( concat ( ?givenName, " ", ?familyName, " ", ?suffix ) AS ?output )
} 

## Select all persons who are born in the month November. Create one output column with strings of the form “John Kennedy (born 25 November 1960)”
#
SELECT ?output
WHERE {
  ?person schema:givenName ?givenName ;
          schema:familyName ?familyName ;
          schema:birthDate ?date ;
  .
  FILTER ( month(?date) = 11 ) 
  BIND ( concat ( 
      ?givenName, " ", ?familyName, " ",
      " (born ", str(day(?date)), 
      " November ", str(year(?date)), ")"
  ) AS ?output )
} 

## Create a list of all persons that have a birthdate specified, and display in the second column the date that the person turned 18 years old.
#
SELECT ?person ?date18
WHERE {
  ?person schema:birthDate ?birth .
  BIND ( concat(
      str(year(?birth) + 18),
      "-",
      str(month(?birth)),
      "-",
      str(day(?birth)))
    AS ?date18AsString )
  BIND ( strDT(?date18AsString, xsd:date) AS ?date18 )
} 

